Apollo 18: The KeY to the Next Mission to the Moon

Jon Xia (jyxia@andrew.cmu.edu) Richard Zhu (rzhu1@andrew.cmu.edu)

1 Abstract

In this project we use a three part model to prove the safety of travelling to the Moon via Hohmann transfer. The first part of the model describes the motion of the spacecraft as it is orbiting the Earth. The second part of the model demonstrates three phases of rocket burns. The first burn puts the spacecraft on an elliptical trajectory to the target orbit. The second burn is fired at the apogee, which circles out the elliptical orbit. Then the Moon and spacecraft orbit in concentric circular orbits until the spacecraft enters the sphere of influence of the Moon, at which point it fires up the third boosting stage, and achieves circular orbit velocity around the Moon. At phase 1, our safety and efficiency condition is that we are on a circular orbit around Earth at a pre-defined radius. The safety and efficiency condition for the second phase is that it stays on the ellipse defined by the appropriate Hohmann transfer. Staying on this ellipse ensures that the spacecraft is on its way to the Moon (efficiency), and that it has not crashed into either the Earth or the Moon (safety). For the final phase, our safety and efficiency condition is that we remain on the pre-defined circular orbit around the Moon. KeYmeraX is able to prove each step of our model, and thus we show that a trans-lunar injection via Hohmann transfer is a viable and safe way of getting spacecraft into lunar orbit. This model is generalizable to other planetary systems.

2 Introduction

After Apollo 17 left the surface of the Moon in 1972, mankind has not attempted another manned mission to the Moon. The goal of this project is to model and prove the safety of a trans-lunar injection from low-Earth orbit to a lunar orbit. With the recent advances from private companies such as SpaceX, Virgin Galactic, and Blue Origin, as well as established space agencies such as the National Aeronautics and Space Administration (NASA) and the European Space Agency (ESA), public interest in space flight is waxing. We believe that the math and calculations detailed in this project are applicable to other space missions in the future, such as a manned inter-planetary mission to Mars.

In space, there are many different ways of getting from point A to point B, and they all have their advantages and disadvantages. Figure 1 shows some of the most relevant types of transfer orbits that we could use to get to a body in our universe. The Hohmann transfer is the one that we will be investigating in this paper, and it consists of two burns, which are detailed in Figure 4. The Hohmann transfer is very simple and very efficient. There is also the direct transfer orbit, which reduces the amount of time it takes to reach the target. However, the drawback to the direct transfer orbit is that it takes much more fuel than the Hohmann transfer orbit. Finally, another viable transfer orbit is the spiral transfer orbit. The drawback of this orbit transfer is that it takes an exorbitant amount of time, but this is the transfer orbit that must be used by spacecraft which use ion thrusters which have a very high specific impulse and efficiency, but can only produce a minute amount of thrust compared to traditional rocket boosters. It is hypothesized that the future of interstellar space travel will consist of traditional rockets bringing these ion rockets into low-Earth orbit, and then using the weak but continuous thrust generated by the ion thrusters to achieve speeds upwards of $0.01c$ to explore the universe. However, our project is more concerned with the immediate exploration of our solar system and more specifically, our moon, which will be done using traditional rockets, so we will use Hohmann transfers in our project to model a spacecraft which starts in low-Earth orbit and then transfers into a higher orbit to then inject into a lunar orbit.



Figure 1: The three most useful and viable transfer orbits

3 Model Overview

We will model the trans-lunar injection by splitting it up into three component parts. The first part will model our spacecraft in a steady-state circular prograde orbit around the Earth at radius r_1. The Moon will also be in a circular prograde orbit around the Earth at radius r_2. When seen from the north pole, the Earth is seen as rotating counter-clockwise, so because the spacecraft and the Moon both revolve in the same direction that the Earth rotates, they will be orbiting counter-clockwise in our model. To the layman, it seems arbitrary that we picked a prograde orbit for our spacecraft rather than a retrograde orbit. However, there are very real and obvious benefits to a prograde rather than retrograde orbit. When launching a spacecraft, NASA generally launches from the east coast of the United States, and then have the spacecraft travel out over the Atlantic ocean. This is because the surface of the Earth is moving relative to stationary at around 460 m/s, or 1000 mph. By launching spacecraft east, into a prograde orbit, NASA (and our model), are able to harness this rotation speed of the Earth. For reference, the International Space Station (ISS), is currently orbiting the Earth at a height of 400km at a velocity of 7600 m/s (17000 mph) relative to the core of the Earth. By launching in a prograde orbit, the Δv required is just 7140 m/s (16000 mph) as opposed to 8060 m/s (18000 mph) required in the opposite direction, which is a 13% increase. By using the Tsiolkovsky rocket equation (Figure 2), a 13% increase in Δv requires a corresponding increase of over 43% in fuel. The reason that we require a proportionally much higher amount of fuel compared to our increase in Δv is because we need more fuel, so in turn that is more fuel that we have to accelerate up to speed. Generally, there are three ways that a spacecraft can increase its Δv. These are staging, increasing specific impulse, or improving propellant mass fraction. Staging is simply where the rocket is put into multiple stages and parts of the rocket are jettisoned after the fuel contained in them has been used. The Apollo missions used this and were split into three stages. The first and second stages, which were the majority of the Saturn V's size and weight, were used just to get the third stage into low-Earth orbit. The third stage was used to get into lunar orbit and then back to Earth. Increasing specific impulse put simply means having better engines. For example, the 5 F1 engines, which were the largest ever man-made rocket engines, of the Saturn V stage 1 provided 7.5 million pounds of thrust, but the space shuttle, which was made decades later, had three main engines and two boosters totalling 7.8 million pounds of thrust. Years of technological improvement was able to provide the space shuttle with far better engines than available on the Saturn V. The last way to increase Δv is just to carry more fuel. Of course, the parameters we have chosen in our model are specific to the Earth-Moon system, but for different star systems, these parameters can be changed, and the model will still be applicable.

$$\Delta v = v_e ln(\frac{m_0}{m_f})$$

Figure 2: Tsiolkovsky rocket equation, where Δv is the maximum change in velocity, v_e is the exhaust velocity, m_0 is the initial mass, m_f is the payload mass, and ln is natural log. The v_e values for the Saturn V's F1 engines were 2600 m/s.

The second part of our model contains three booster burns, of which the first two comprise a Hohmann transfer from low-Earth orbit to our desired orbit (Figure 4). We want the spacecraft to orbit the Moon at a radius r_3, so the desired orbit distance from the Earth should be $r_2 - r_3$. The last rocket burn is to attain orbit velocity around the Moon to ensure that our spacecraft does not crash into it after entering its sphere of influence. Because we wanted the spacecraft to be in a circular orbit, the speed of the spacecraft relative to the Moon must be $\sqrt{Gm_{moon}/r_3}$, which is calculated by equating the centripetal force $F_c = mv^2/r$ to the gravitational force, $F_g = GMm/r^2$ (Figure 3). In all of these equations and in the rest of this paper, G is Newton's gravitational constant, $6.674 * 10^{-11} m^3 kg^{-1} s^{-2}$.

$$\frac{m_{craft} v^2}{r_3} = \frac{G m_{craft} m_{moon}}{r_3^2}$$
$$v^2 = \frac{G m_{moon}}{r_3}$$
$$v = \sqrt{\frac{G m_{moon}}{r_3}}$$

Figure 3: Derivation of orbit velocity necessary to maintain circular orbit around Moon

The final part of our model is simply the spacecraft in a steady state retrograde orbit around the Moon. The reason that a retrograde orbit was chosen as opposed to a prograde orbit is because entering a retrograde orbit allows us to transfer to a lower orbit than would be required by a prograde orbit (Figure 5). This lower Hohmann transfer helps to conserve fuel as well as ensure that we don't enter the Moon's sphere of influence while we are still transferring into our desired orbit.



Figure 4: Diagram showing the Hohmann transfer to achieve a higher circular orbit from a lower circular orbit. In this diagram, $r_4 = r_2 - r_3$

4 Model Limitations

Due to the astronomical complexities of space-flight, there are many assumptions and simplifications that we made to our model. Perhaps the biggest approximation that we do is using the method of patched conics. In space, everything exhibits a gravitational force upon everything else. However, by using patched conics, we ignore the effects of all but one body's gravitational pull while we are in its sphere of influence. Technically, everything in the Solar System is under the influence of the Sun's gravity, but when we are orbiting close to the Earth, we are in its sphere of influence, and the method of patched conics allows us to just consider the Earth's gravity, while ignoring that of the Sun and that of the Moon. Once we get closer to the Moon, and enter its sphere of influence, then we ignore the gravitation effects of the Earth and just treat the system as two bodies, the Moon and our spacecraft.



Figure 5: Diagram demonstrating how a retrograde orbit around the Moon (green) requires a lower Hohmann transfer than a prograde orbit (red).

Another approximation that we had to make in order to reduce computational complexity was to make the mass of our spacecraft 0. This is done because relative to the masses of the Earth and the Moon, which are on the order of 10^24kg and 10^22kg respectively, the mass of the spacecraft is many millions of times lower, so effectively it can be approximated as 0. This approximation allows us to ignore any effects of gravity imparted on the Moon and Earth caused by our spacecraft. A nice side effect of this approximation is that it also allows us to have instantaneous and unbounded acceleration of our vehicle. Hohmann transfers require instantaneous acceleration, so this approximation is necessary if we wish to use this orbit transfer technique. In real life, the rocket boosters would need to be fired for a non-zero amount of time in order to achieve the Δv required. For a frame of reference, in the actual Apollo missions, the trans-lunar injection rocket burn lasted for about 350 seconds, providing a total Δv of about 3000 m/s. Because of this, true Hohmann transfers are not used in real life, but the Apollo missions did use a modified Hohmann transfer. By integrating over the path taken by the spacecraft, it is calculated that it would take over four days to reach the Moon from Earth using a true Hohmann transfer. However, in actuality, the Apollo missions took around three days. That is because NASA wanted to ensure that in the event of an engine failure, the Apollo capsule would be on a trajectory such that it would be gravitationally slingshotted back to Earth if anything were to go awry. That being said, we use a Hohmann transfer in our model for its simplicity and efficiency.

Another aspect of our model which is unlike real life is the fact that we don't time our Hohmann transfer. The Apollo missions were exactly timed such that they launched at the perfect time to have their trajectory be straight to the Moon. Our model, however, is a bit more loose. We can start our Hohmann transfer at any time, such that when we arrive at our target orbit, the Moon could be on the other side of the Earth. However, this is no problem. The spacecraft is orbiting the Earth on a circle of radius of $r_2 - r_3$, and the Moon is orbiting the Earth on a circle of radius r_2. According to Kepler's Third Law, the period of rotation of an object around another is proportional to the square root of the cube of the semi-major axis. Because we are dealing with circles, this is just the radius. Since we have the spacecraft and the Moon orbiting the Earth on different radii, they also have different periods, and since they are both travelling in prograde orbits, the spacecraft is bound to come into the sphere of influence of the Moon eventually.

One of the less significant approximations that we made was approximating the orbits of the Moon and the spacecraft as circles. The true eccentricity of the Moon's orbit is 0.0549. This approximation was made to simplify the proofs in KeYmaeraX.

5 Proving with KeYmaeraX

To prove the validity of the three parts of our model, we used KeYmaeraX, an axiomatic tactical theorem prover for hybrid systems developed by Professor Andre Platzer's research group at Carnegie Mellon University. First we formalized each of our three sub-models as a hybrid program with initial conditions, continuous dynamics, discrete control, and safety and efficiency conditions. If, given all specified initial conditions, a sub-model's continuous dynamics satisfy the specified safety and efficiency conditions while subject to discrete control, then our sub-model will have been proven to be valid. When all three sub-models have been proven valid, we can say that the entire model has been proven valid.

Our first sub-model establishes our spacecraft and the Moon simultaneously orbiting the Earth. As such, it's initial conditions are that for a given orbit radius s_r, the spacecraft is s_r away from the center of the Earth and is travelling tangent to that orbit. The same is true for the moon at it's orbit radius of m_r. In addition, both orbits must be greater than the Earth's non-zero radius. Furthermore, m_r is greater than s_r and the distance between the two orbits is greater than the radius of the moon. The continuous dynamics of the system are that of 2 bodies going in a circle around a single point, the Earth. For computation simplicity, we set that point to be the origin. The differential equations of circular motion around the origin are $x' = rv\partial x, y' = rv\partial y, \partial x' = -v\partial y, \partial y' = v\partial x$ where x' is motion in the x-direction, y' is motion in the y-direction, v is the magnitude of the linear velocity, and r is the radius of the circle. The differential equations representing the continuous dynamics of the first sub-model are then just two sets of circular dynamics. The safety and efficiency condition is that the no matter how long the differential equations evolve, the two bodies stay on their orbits. Finally, there is no discrete control for this sub-model. Proving the sub-model then boils down to proving that the circle dynamics described above preserve circular motion. This can be done using a combination of the differential cut, differential invariant, and differential weakening sequents:

Differential Cut:

$$\text{dC}\ \frac{\Gamma \vdash [x' = f(x) \& Q]C, Delta \qquad \Gamma \vdash [x' = f(x) \& Q \wedge C]P, \Delta}{\Gamma \vdash [x' = f(x) \& Q]P, \Delta}$$

Differential Invariant:

$$\text{dC}\ \frac{\Gamma, Q \vdash P, \Delta \qquad Q \vdash [x' := f(x)](P)'}{\Gamma \vdash [x' = f(x) \& Q]P, \Delta}$$

Differential Weakening:

$$\text{dW}\ \frac{\Gamma_{const}, Q \vdash P, \Delta_{const}}{\Gamma \vdash [x' = f(x) \& Q]P, \Delta}$$

For details, see the Appendix for the full KeYmaeraX proof.

Our second sub-model assumes that our spacecraft is in orbit around Earth and performs a Hohmann transform to move the spacecraft off it's circular Earth orbit onto an elliptical Hohmann transfer orbit. Using Figure 4 above, the spacecraft starts on it's circular Earth orbit with a velocity of $\sqrt{\frac{GM}{R_1}}$ where G is the gravitational constant, M is the mass of the Earth-spacecraft system (but the assume the mass of the spacecraft is negligible compared to the Earth so really M is the mass of the Earth), and R_1 is the radius of the orbit. It then performs an instantaneous velocity change corresponding to Δv to $\sqrt{GM\left(\frac{2}{r}-\frac{1}{a}\right)}$, which is the velocity needed to maintain an elliptical Hohmann transfer orbit with semi-major axis $a = \frac{R_4+R_1}{2}$ in Figure 4. GM is the same as before and r is the distance between the Earth and the spacecraft, which is no longer constant. This Hohmann transfer will take the spacecraft from it's original Earth orbit with radius R_1 to the second orbit with radius R_4. The initial conditions of the second sub-model include $R_4 > R_1$, $R_{moonOrbit} > R_4$, and $R_{moonOrbit} - R_4 < R_{moonSOI}$, where the moon's SOI (sphere of influence) is the area under which the Moon's gravitational field attracts objects (as we are using the patched conics approximation). This ensures that at the end of the Hohmann transfer, the spacecraft will be on an orbit close to the Moon and can be picked up by it's gravitational field and enter into a Lunar orbit. Note that these initial conditions are either not affected by or follow from the safety and efficiency conditions of the first sub-model, which ensures that we can transition from the first sub-model to the second. The continuous dynamics of the second sub-model is that of an ellipse and can be modeled by the differential equations: $x' = v_x, y' = v_y, v_x' = \frac{-GMr_x}{r^3}, vy' = \frac{-GMr_y}{r^3}$ where G, M, r are still the gravitational constant, the mass of the earth, and the distance between the Earth and the spacecraft. r_x and r_y are the x and y components of the distance r so $r_x^2 + r_y^2 = r^2$. The safety and efficiency condition is that at all times, the spacecraft's velocity after the initial Δv is consistent with being on an ellipse: $v_x^2 + v_y^2 = GM\left(\frac{2}{r}-\frac{1}{a}\right)$. Note that for mathematical simplicity, we centered the ellipse at (0,0) while placing the Earth at $(R_1 - a, 0)$ and the starting position of the spacecraft at $(-a, 0)$. This is simply a change in coordinate system and so can be done regardless of where the spacecraft ended up relative to the Earth at the end of the first sub-model. The proof of this sub-model mainly relies on the differential invariant rule and can be found in the Appendix under Model Part 2.

The third and final sub-model assumes that our spacecraft is in a Hohmann transfer elliptical orbit and has reached the point of the orbit farthest from the Earth. This places it within the sphere of influence of the Moon and so a second Δv is done to change the velocity of the spacecraft from $\sqrt{GM\left(\frac{2}{r}-\frac{1}{a}\right)}$ to $\sqrt{\frac{Gm}{r}}$ where m is the mass of the Moon. This pushes the spacecraft off it's elliptical orbit onto a circular Lunar orbit with radius r. r is initialized to be greater than the radius of the Moon but less than or equal to it's sphere of influence. It is assumed that we chose an appropriate launch window such that the Moon will be there to "catch" the spacecraft at the end of its Hohmann transform. However, even if that wasn't the case, we could have changed the velocity to $\sqrt{\frac{GM}{R_4}}$ where M is the mass of the Earth and R_4 is the second orbit in Figure 4, in which case the spacecraft would orbit the Earth until it caught up to the Moon or vice-versa, in which case another Δv could be performed to get into lunar orbit. The continuous dynamics of this hybrid program are the same as for the first sub-model and so will not be repeated, as our the safety and efficiency conditions. The detailed model and proof can be found in the Appendix under Model Part 3.

6 Conclusion

Through KeYmaeraX, because all three of our models have proven, we have shown that we can safely and efficiently get into lunar orbit from low-Earth orbit by using a Hohmann transfer. However, this model is not just limited to the Earth and the Moon. We could modify it to a trip to Mars by simply replacing the Earth with the Sun and the Moon with Mars. Because all the objects in the universe consist of objects revolving around one another, there are unlimited applications for this model in future exploration of our universe.

7 References

[1] Lutze, F. (2003, August). Summary of Patched Conic Approximations. Retrieved April 2017, from Virginia Tech.

[2] R. Biesbroek, G. J. (2000, August). Ways to the Moon? Retrieved April 2017, from European Space Agency: http://www.esa.int/esapub/bulletin/bullet103/biesbroek103.pdf

[3] S. Widnall, J. P. (2008, Fall). Orbit Transfers and Interplanetary Trajectories . Retrieved April 2017, from Open Course Ware - MIT.

[4] Tom Logsdon, Orbital Mechanics: theory and applications, John Wiley & Sons, 1998

[5] К. Ціолковскій, Изслдованіе мировыхъ пространствъ реактивными приборами, 1903

[6] J. Spurmann, (2010, February). Lunar Transfer Trajectories. Retrieved April 2017, from ftp://dim13.org/pub/doc/TN$_1002.pdf$

[7] W. Hohmann, The Attainability of Heavenly Bodies (Washington: NASA Technical Translation F-44, 1960) Internet Archive

[8] R. W. Farquhar (Jun 1972). "A Halo-Orbit Lunar Station". Astronautics Aeronautics.

[9] V. Szebehely, Theory of Orbits, The Restricted Problem of Three Bodies, Yale University, Academic Press, 1967.

[10] Vallado, David Anthony (2001). Fundamentals of Astrodynamics and Applications. Springer. p. 317.

[11] Forward, Robert L. "A Transparent Derivation of the Relativistic Rocket Equation"

[12] Zubrin, Robert (1999). Entering Space: Creating a Spacefaring Civilization. Tarcher/Putnam.

8 Appendix

Model Part 1, Earth Orbits: KeYmaeraX Model and Proof

```
ArchiveEntry "earth_orbits".

Functions.
R em. /* earth mass */
R ex. /* earth x */
R ey. /* earth y */
R er. /* earth radius */
R sr. /* spacecraft orbit radius */
R sv. /* spacecraft linear velocity */
R mo. /* moon orbit radius */
R mr. /* moon radius */
R mv. /* moon linear velocity */
R G.  /* gravitational constant */
End.

ProgramVariables.
R sx. /* spacecraft x */
R sy. /* spacecraft y */
R sdx./* spacecraft x' direction */
R sdy./* spacecraft y' direction */
R mx. /* moon x */
R my. /* moon y */
R mdx./* moon x' direction */
R mdy./* moon y' direction */
End.

Problem.
(
    G = 6.67 * 10 ^-11 & /* set gravitational constant */
    ex = 0 & ey = 0 & /* center coordinate system on earth */
    em > 0 & /* earth mass > 0 */
    /* set spacecraft in orbit around earth */
    sx^2 + sy^2 = sr^2 & sv = ((G * em) / sr)^0.5 & sdx^2 + sdy^2 = 1 &
    sx = sr * sdy & sy = -sr * sdx &
    /* set moon in orbit around earth */
    mx^2 + my^2 = mo^2 & mv = ((G * em) / mo)^0.5 & mdx^2 + mdy^2 = 1 &
    mx = mo * mdy & my = -mo * mdx &
    /* moon orbit radius > spacecraft orbit radius > earth radius > 0 */
    mo > sr & sr > er & er > 0 &
    mo - sr > mr /* distance between spacecraft and moon orbits > moon radius */
)/* Initial Conditions */
->
[
    {
        /* spacecraft orbit kinematics */
        sx' = sr * sv * sdx, sy' = sr * sv * sdy,
        sdx' = -sv * sdy, sdy' = sv * sdx,
        /* moon orbit kinematics */
        mx' = mo * mv * mdx, my' = mo * mv * mdy,
        mdx' = -mv * mdy, mdy' = mv * mdx
    }
]
(
    sx^2 + sy^2 = sr^2 & sdx^2 + sdy^2 = 1 & /* spacecraft stays in orbit */
    mx^2 + my^2 = mo^2 & mdx^2 + mdy^2 = 1 /* moon stays in orbit */
) /* Safety and Efficiency Condition */
```

```
End.

Tactic "earth_orbits: Proof 1".
  implyR('R) ; boxAnd('R) ; andR('R) ; <(
  dC({`sdx^2+sdy^2=1`}, 'R) ; <(
    dC({`sx=sr()*sdy&sy=-sr()*sdx`}, 'R) ; <(
      dW('R) ; master,
      dI('R)
      ),
    dI('R)
    ),
  boxAnd('R) ; andR('R) ; <(
    dI('R),
    boxAnd('R) ; andR('R) ; <(
      dC({`mx=mo()*mdy&my=-mo()*mdx`}, 'R) ; <(
        dI('R),
        dI('R)
        ),
      dI('R)
      )
    )
  )
End.

End.
```

Model Part 2, Hohmann Transform: KeYmaeraX Model and Proof

```
ArchiveEntry "hohmann_transfer".

Functions.
R em. /* earth mass */
R ex. /* earth x */
R ey. /* earth y */
R er. /* earth radius */
R r1. /* initial spacecraft orbit radius */
R r2. /* final spacecraft orbit radius */
R mo. /* moon orbit radius */
R mr. /* moon radius */
R ms. /* moon sphere of influence radius */
R a.  /* Hohmann transfer (ellipse) semi-major axis */
R G.  /* gravitational constant */
End.

ProgramVariables.
R x.  /* spacecraft x */
R y.  /* spacecraft y */
R vx. /* spacecraft x velocity */
R vy. /* spacecraft y velocity */
R d.  /* distance between spacecraft and earth */
End.

Problem.
(
    /* gravitational constant */
    G = 6.67 * 10 ^-11 &
    /* place earth at (r1 - a, 0) */
    ex = r1 - a & ey = 0 &
    /* place spacecraft at (-a, 0) */
    x = -a & y = 0 &
```

```
    /* set spacecraft velocity so that it's in orbit around earth */
    vx = 0 & vy = -(G * em / r1)^0.5 &
    a = (r1 + r2) / 2 & /* compute the ellipse semi-major axis */
    /* moon orbit > final spacecraft orbit > initial spacecraft orbit > */
    /* earth radius > 0, earth mass > 0 */
    mo > r2 & r2 > r1 & r1 > er & er > 0 & em > 0 &
    /* moon orbit is more than moon radius apart from final spacecraft orbit */
    mo - r2 > mr &
    /* final spacecraft orbit intersects moon's sphere of influence */
    mo - r2 < ms &
    /* moon's sphere of influence doesn't extend to initial spacecraft orbit */
    mo - r2 > r1 &
    /* moon's orbit > moon's sphere of influence > moon radius > 0 */
    mo > ms & ms > mr & mr > 0
)/* Initial Conditions */
->
[
    /* change velocity to switch from circular earth orbit to elliptical */
    /*   hohmann transfer orbit */
    vy := -(G * em * (2 / ((x-ex)^2+y^2)^0.5 - 1 / a))^0.5;
    {
        /* spacecraft hohmann transfer kinematics */
        x' = vx, y' = vy,
        vx' = -(x - ex) * G * em / ((x-ex)^2+y^2)^1.5,
        vy' = -y * G * em / ((x-ex)^2+y^2)^1.5 &
        /* exit this model when we reach lunar orbit so we can switch to */
        /*   the lunar orbit model */
        x <= a
    }
]
(
    /* velocities consistent with being on elliptical orbit around earth */
    vx^2 + vy^2 = G * em * (2 / ((x-ex)^2+y^2)^0.5 - 1 / a)
) /* Safety and Efficiency Condition */
End.

Tactic "hohmann_transfer: Proof 1".
  implyR('R) ; composeb('R) ; assignb('R) ; dI('R)
End.

End.
```

Model Part 3, Lunar Orbit: KeYmaeraX Model and Proof

```
ArchiveEntry "lunar_orbit".

Functions.
R em. /* earth mass */
R er. /* earth radius */
R mm. /* moon mass */
R mx. /* moon x */
R my. /* moon y */
R mo. /* moon orbit radius */
R mr. /* moon radius */
R ms. /* moon sphere of influence radius */
R r1. /* initial spacecraft earth orbit radius */
R r2. /* final spacecraft earth orbit radius */
R r3. /* space craft lunar orbit radius */
R a.  /* Hohmann transfer (ellipse) semi-major axis */
R G.  /* gravitational constant */
```

```
End.

ProgramVariables.
R x.  /* spacecraft x */
R y.  /* spacecraft y */
R dx. /* spacecraft x' direction */
R dy. /* spacecraft y' direction */
R v.  /* spacecraft linear velocity */
End.

Problem.
(
    G = 6.67 * 10 ^-11 & /* gravitational constant */
    mx = 0 & my = 0 & /* center coordinate system on moon */
    a = (r1 + r2)/2 & /* set hohmann transfer semi-major axis */
    x^2 + y^2 = r3^2 & /* spacecraft is travelling tangent to lunar orbit */
    dx^2 + dy^2 = 1 &  /*    r3 distance from center of moon */
    v = G * em * (2/r2 - 1/a) & /* spacecraft's velocity is velocity at end */
                                /*    of hohmann transfer */
    x = r3 * dy & y = -r3 * dx & /* relating x, dx, y, dy */
    em > mm & mm > 0 & /* Earth mass > Moon mass > 0 */
    mo > r2 & /* moon orbit > final spacecraft earth orbit */
    r2 > r1 & /*    > initial spacecraft earth orbit */
    r1 > er & /*    > earth radius */
    er > 0 &  /*    > 0 */
    ms >= r3 & /* moon sphere of influence >= spacecraft moon orbit */
    r3 > mr &  /*    > moon radius */
    mr > 0 &   /*    > 0 */
    ms - r2 > r1 /* spacecraft initial orbit not in moon sphere of influence */
)/* Initial Conditions */
->
[
    v := ((G * mm) / r3)^0.5; /* change velocity to switch from hohmann */
                              /*    transfer orbit to lunar orbit */
    {
        x' = r3 * v * dx, y' = r3 * v * dy, /* spacecraft lunar orbit dynamics */
        dx' = -v * dy, dy' = v * dx
    }
]
(
    x^2 + y^2 = r3^2 & dx^2 + dy^2 = 1 /* spacecraft stays in lunar orbit */
) /* Safety and Efficiency Condition */
End.

Tactic "lunar_orbit: Proof 1".
  implyR('R) ; composeb('R) ; assignb('R) ; dC({`dx^2+dy^2=1`}, 'R) ; <(
  boxAnd('R) ; andR('R) ; <(
    dC({`x=r3()*dy&y=-r3()*dx`}, 'R) ; <(
      dW('R) ; master,
      dI('R)
      ),
    dI('R)
    ),
  dI('R)
  )
End.

End.
```